
Lecture 37:

• Intro to Answer-Set Programming (ASP)

Announcements:

• HW-8 out

• Last (make-up) quiz on Friday

Overview

Recall:

• Propositional logic: p ∧ ¬q → r ... r ← p ∧ ¬q
• Predicate logic: p(x, y) ∧ ¬q(y)→ r(x) ... r(x)← p(x, y) ∧ ¬q(y)

Basic Idea of Logic Programming (LP):

• Use logic-based syntax and semantics as a computational model(s)

• Programs as sets of “facts” (data) and “rules” (possibly recursive)

Examples of LP in practice:

• SQL (although not immediately obvious)

• Datalog (DB query language, focus on recursion)

• Prolog (one of the first mainstream LP languages)

• Answer Set Programming (disjunctive Datalog as a PL) ... DLV, Clingo

• Many new systems (declarative data science, neurosymbolic AI, etc.)

Overview

Potassco Answer-Set Programming (ASP) tools: ... see potassco.org

• we’ll look at clingo ... but various other tools

• download using conda, homebrew, ubuntu apt, etc.

• can use within python (clingo api)

• can use via web at potassco.org/clingo/run/ or ASP chef

Designed for solving combinatorial optimization problems

• find optimal (min/max/best) solution from (finitely) many possibilities

• often NP-Complete problems, e.g., vehicle routing, scheduling, games, ...

ASP programs expressed in Guess-and-Check / Generate-and-Check style

1. Select and transform input data ... query-like

2. Generate possible solutions ... “guess” phase

3. Filter those that satisfy certain solution properties ... “check” phase

Overview

Exercise: sketch an algorithm that:

• takes a graph as input (with nodes and edges)

• assigns colors (R, G, B) to nodes where no adjacent nodes have same color

A (naive) “greedy” approach: ... note 3-colorability is NP-Complete!

1: For each node x1 . . . xn in order
2: set color(xi) based on color of xi’s neighbors
3: if no possible color for xi, backtrack to xi−1

4: if backtrack to x1, graph is not 3-colorable

Overview

3-coloring using Answer Set Programming clingo

1 % assume: graph edges given , grab just the nodes

2 node(X) :- edge(X,_).

3 node(X) :- edge(_,X).

4 % guess: color for each node (better to use a choice rule)

5 color(X,red) | color(X,green) | color(X,blue) :- node(X).

6 % constraint: adjacent colors

7 :- edge(X,Y), color(X,C), color(Y,C).

To run: clingo 3-color.lp 3-color-graph.lp ...

Note: This program can also find all possible colorings!

• clingo 3-color.lp 3-color-graph.lp 0 ... all answers

Overview

Exercise: Sketch algorithm for xkcd.com/287/ given any menu and total price

Overview

1 total (1505). % total desired cost (using ints)

2 amount (0..7). % amount (0). amount (1). etc.

3 % the appetizer menu

4 item_price(fruit ,215). item_price(fries ,275).

5 item_price(salad ,335). item_price(wings ,355).

6 item_price(stick ,420). item_price(plate ,580).

7 % grab the items

8 item(F) :- item_price(F,_).

9 % guess an amount to order (" choice" rule)

10 1 = { order(F,N) : amount(N) } :- item(F).

11 % calculate the item totals

12 total_order(F,N,T) :- order(F,N), item_price(F,A), T = N * A.

13 % constrain to be 1505 total

14 :- total(V), V != #sum { T,F : total_order(F,N,T) }.

15 % only show the total_order relation

16 #show total_order /3.

Clingo Basics

Consider propositional statements:

p ... proposition p is true

q ← p ... if p is true, q is true

Q: What do we know to be true as a result of these statements?

1. p is true (given as a “fact”)

2. since p is true, q must also be true (inferred from the rule)

{p, q} is a program answer ... aka answer set, stable model, possible world

In clingo we write:

1 p. % p is a fact

2 q :- p. % same as p -> q

